
Subject: Random Access Text  (like getline)
Posted by lefsky on Tue, 22 Jun 2004 22:03:29 GMT
View Forum Message <> Reply to Message

I am looking for way to randomly access a text file, as getline will
do in python. Essentially, getline will allow you to access any line
in a text file by giving it the line number in the file (e.g.
getline(100)). I can't see a way to do that with a text file in
IDL....

Subject: Antw: Random Access Text (like getline)
Posted by Harald Jeszenszky on Thu, 24 Jun 2004 07:23:47 GMT
View Forum Message <> Reply to Message

Hi Michael,

The following code reads consecutive blocks of text lines until the choosen
text line number is contained in the block. To change the performance you
can fiddle around with the buffer size (variable bufSize) but the given
value of 100 is a good starting point.

Regards,
Harald

 ;----------------------------------------------------------- ------------

FUNCTION getline, file, lineNum, ERROR=error

    CATCH, error
    IF (error NE 0) THEN BEGIN
    	MESSAGE, !ERROR_STATE.MSG, /CONTINUE
    	IF (N_ELEMENTS(unit) NE 0) THEN $
    	    FREE_LUN, unit
    	RETURN, ''
    ENDIF
    
    ; check parameters
    
    IF (N_PARAMS() NE 2) THEN $
    	MESSAGE, 'Missing parameter.', /NONAME
    	
    lineNum = LONG(lineNum)
    IF (lineNum LT 1) THEN $
    	MESSAGE, 'Line number starts from 1.', /NONAME
    	
    ; read blocks of text lines

    bufSize = 100L
    lineBuf = STRARR(bufSize)
    lineCnt = 0L

    OPENR, unit, file, /GET_LUN
    ON_IOERROR, END_OF_LOOP
    WHILE (NOT EOF(unit)) AND (lineCnt LT lineNum) DO BEGIN
    	READF, unit, lineBuf
    END_OF_LOOP:
    	lineCnt = lineCnt + (FSTAT(unit)).TRANSFER_COUNT
    ENDWHILE
    ON_IOERROR, NULL
    FREE_LUN, unit
    
    ; return text line
    
    IF (lineCnt LT lineNum) THEN $
    	MESSAGE, 'End of file reached.', /NONAME
    	
    RETURN, lineBuf[(lineNum - 1) MOD bufSize]
    
END ; getline

 ;----------------------------------------------------------- ------------

>>>  Michael Lefsky<lefsky@cnr.colostate.edu> 23.06.04 00:03:29 >>>
I am looking for way to randomly access a text file, as getline will
do in python. Essentially, getline will allow you to access any line
in a text file by giving it the line number in the file (e.g.
getline(100)). I can't see a way to do that with a text file in
IDL....

